# $Id: PKGBUILD 87585 2013-04-03 13:22:20Z stativ $
# Maintainer: Lukas Jirkovsky <l.jirkovsky@gmail.com>
pkgname=luxblend25
pkgver=1.2.1
_pkgver=c9deae0646d0
_blender=2.66
pkgrel=1
pkgdesc="A LuxRender exporter for Blender."
arch=('any')
url="http://www.luxrender.net/"
license=('GPL')
depends=('blender' 'luxrender')
source=(https://bitbucket.org/luxrender/luxblend25/get/$_pkgver.tar.bz2)
md5sums=('a8d78362318a1f0b795c130638a2f4b1')

package() {
  install -d -m755 "$pkgdir"/usr/share/blender/$_blender/scripts/addons
  cp -a "$srcdir"/luxrender-luxblend25-$_pkgver/src/luxrender \
    "$pkgdir"/usr/share/blender/$_blender/scripts/addons
  # change the search path in exporter so it finds pylux in its new location
  sed -i 's|from.*import pylux|import pylux|' "$pkgdir/usr/share/blender/$_blender/scripts/addons/luxrender/outputs/pure_api.py"
}

# vim:set ts=2 sw=2 et:
